

CORSO DI EXCEL PER UFFICIO

Contenuto

Il foglio di calcolo	2	Subtotali in una tabella	25
Introduzione	2	Ordinamento di record	25
I riferimenti	3	Calcolo dei subtotali	26
Le formule e le funzioni	4	Formattazione dei subtotali e stili	28
Le formule	4	I grafici	29
Le funzioni	5	Stampa ed esportazione dei grafici	34
Il foglio di esempio	7	Stampa di uno o più grafici all'interno del fogli di lavoro	34
Riferimenti assoluti e relativi	10	Stampa di un grafico su un foglio a sé	34
L'esempio dell'elenco di libri	13	Copia e incolla di grafici da Excel a Word	35
Aggiunta di campi calcolati	13	Nomi al posto di riferimenti	36
Estrazione di informazioni di riepilogo	17	Ulteriori note (intersezioni di nomi, nomi predefiniti, nomi locali)	38
Funzionalità di Excel	19	Un esempio di funzioni nidificate	40
Spostamento tra le celle	19	Suggerimenti per un migliore uso di Excel	43
Selezione di celle	19	Tasti di scelta rapida	43
Spostamento e copia di celle	19	Opzioni utili	44
Copia elementare delle celle	20		
Spostamento elementare delle celle	20		
Riferimenti per le celle copiate	20		
Spostamento e copia con il mouse	21		
"Incolla speciale" per incollare solo alcuni aspetti di una cella	22		

IL FOGLIO DI CALCOLO

INTRODUZIONE

Un "**foglio di calcolo**" è un oggetto di un programma per computer costituito da un insieme di celle, organizzate in righe e colonne, atte a memorizzare dati ed effettuare operazioni matematiche. Microsoft Excel è un programma che contiene un foglio di calcolo. In realtà Excel contiene anche altri oggetti – soprattutto un disegnatore di grafici ed un linguaggio di programmazione – ma l'oggetto di più largo utilizzo al suo interno è senza dubbio il foglio di calcolo, così che possiamo ragionevolmente dire che Excel è un foglio di calcolo.

La maglia di **righe** e **colonne** in cui è suddiviso un foglio di calcolo identifica una serie di caselle – le **celle** – che sono le entità elementari in cui si memorizzano dati o con cui si specificano calcoli. È assai buona norma che *ogni cella contenga una sola informazione*. Ad esempio, se devo memorizzare dei nominativi e, per ciascuno di essi, la data di nascita, è in genere vantaggioso porre il nominativo in una cella e la data di nascita in un'altra.

Più in generale, è vantaggioso *non riunire mai nella stessa cella due informazioni che devono essere trattate (anche) separatamente*. Per "trattate" intendo che su di esse si facciano dei conti, ma anche che semplicemente si desideri ordinare un elenco in base ad esse.

La questione dipende anche dal contesto, cioè dall'uso che prevedo di fare delle informazioni. Un esempio tipico: nome e cognome vanno nella stessa cella se li voglio considerare come un'unica entità. Se invece devo ad esempio fare delle statistiche sui nomi (contare quante persone si chiamano "Mario"), devo necessariamente usare due celle. In più, se le celle contengono, anche legittimamente, due informazioni distinte, è vantaggioso cominciare sempre da quella gerarchicamente più importante. Se uso una cella per nome e cognome, è bene che scriva prima il cognome, in modo da poter, ad esempio, ordinare un elenco per cognome, che è l'ordinamento più spontaneo (naturalmente se uso due celle distinte, posso ordinare a piacere per nome o cognome).

I RIFERIMENTI

Le **celle** vengono "chiamate" (cioè univocamente identificate) mediante la loro colonna e la loro riga.

Le **colonne** sono identificate da lettere: A, B, C, ... Z, AA, AB, AC, ... AZ, BA, BB... Ci sono 16.384 colonne.

Le **righe** sono identificate da numeri: 1, 2, 3... Ci sono dunque 1.048.576 righe.

La cella che sta sulla prima colonna e sulla prima riga è dunque A1 (prima la lettera, poi il numero, senza spazi, non importano le maiuscole).

Se voglio riferirmi a più celle adiacenti, cioè ad un **intervallo** di celle, le identifico tramite la prima e l'ultima, separandole con i *due punti*. (il "separatore di intervallo"). Se le celle sono poste in verticale su una sola colonna o in orizzontale su una sola riga non c'è problema. Se invece formano un'area rettangolare, la prima è quella in alto a sinistra, l'ultima quella in basso a destra.

Si ha quindi:

- | | |
|---------------------------------------|--------------------------|
| ➤ Una singola cella | A1, C3, AB14, Z1234, ... |
| ➤ Un intervallo di celle incolonnate | A3:A5, AA11:AA1200, ... |
| ➤ Un intervallo di celle allineate | A3:C3, B8:AD8, |
| ➤ Un intervallo di celle rettangolari | A3:D10, C2:AA111, ... |
| ➤ Una singola riga | 1:1, 3:3, ... |
| ➤ Una singola colonna | A:A, BC:BC, ... |
| ➤ L'intero foglio di calcolo | 1:1048576 |

LE FORMULE E LE FUNZIONI

LE FORMULE

All'interno delle celle posso scrivere tre cose:

- del testo;
- dei numeri;
- delle formule.

Il **testo** è tutto quello che contiene almeno un carattere che non sia una cifra numerica o un *separatore* che abbia significato all'interno di un numero. Il più semplice separatore è quello decimale (cioè, in italiano, la "virgola"). Quindi *ciao*, *1A* o *1..* sono dei testi, ma non *1.2* perché (se uso il punto come separatore decimale) esso può essere interpretato come "uno virgola due". Anche *11/5/98* non è un testo, perché può essere interpretato come una data (confronta sotto).

I **numeri** sono tutto quello che può essere interpretato come un numero (e su cui possono essere fatti dei conti). Anche una data o un'ora sono numeri (indipendentemente da come vengono visualizzate, le date sono tutte memorizzate come giorni contati a partire dal 1° gennaio del 1900).

Le formule sono l'argomento della presente sezione, e ne ripareremo fra un momento. Nel frattempo però è utile precisare qualcosa in più.

Excel fa conti sui numeri, ma fa qualcosa di molto simile ai conti anche sui testi. Ad esempio se mi domando quante lettere è lunga la parola "ciao", sto facendo in un certo senso un conto su un testo. Anche quanto confronto due testi (per es. per vedere se sono uguali) sto facendo se non proprio un conto, almeno un'operazione su testi.

Se non specifico diversamente, *i testi sono sempre allineati a sinistra e i numeri a destra nelle celle.*

Se per caso Excel interpreta qualcosa come numero, mentre io voglio che rimanga del testo (può succedere che un mio testo assomigli soprattutto ad un'ora o eventualmente ad una data), *devo far precedere il testo da un apostrofo* (che non verrà visualizzato nella cella).

E veniamo alle **formule**. Le formule sono espressioni (non necessariamente matematiche in senso stretto) che *restituiscono dei valori*. Sia testo, sia numeri sono in generale **valori**. Le formule prendono dei valori (cioè testi o numeri) e restituiscono altri valori, ad essi legati dall'operazione che abbiamo eseguito.

Tutte le formule cominciano con un segno di uguale .

Le formule usano le quattro operazioni matematiche:

somma	+
sottrazione	-
moltiplicazione	*
divisione	/

Gli operandi delle formule possono essere:

➤ **numeri**

=3+5 è una formula che restituisce 8

➤ **riferimenti**

=A1+A3 restituisce il numero contenuto in A1 sommato al numero contenuto in A3.

Si noti che:

- A1 e A3 devono contenere davvero numeri. Se così non è ottengo un errore di valore (#VALORE!).
- L'operazione deve avere significato matematico: =A1/A3 restituisce un errore di divisione per zero (#DIV0!) se A3 è zero (o anche se in A3 non c'è scritto nulla).
- Se cambio il valore di una cella a cui una formula si riferisce, la formula è automaticamente aggiornata. Se in A1 ho 2 e in A3 ho 3, =A1+A3 restituisce 5. Se ora scrivo 4 in A3, =A1+A3 mi passa automaticamente ed istantaneamente a 6. Il *ricalcolo automatico* è tra le potenzialità più grandi di un foglio di calcolo.

➤ **numeri e riferimenti**, mescolati secondo necessità, con più operazioni, anche racchiuse tra le opportune parentesi: =3+(A3-5)/A1 è per esempio una formula valida

LE FUNZIONI

In aggiunta alle operazioni matematiche elementari, Excel mette a disposizione una lunga serie di **funzioni** preconfezionate, che svolgono (nei casi più semplici e comuni) un insieme di operazioni elementari. In casi più sofisticati, le funzioni realizzano compiti non riconducibili ad operazioni elementari (ad esempio il *confronto* tra valori).

Le funzioni hanno un nome che le identifica e una serie di **argomenti** su cui agiscono. Gli argomenti si indicano immediatamente dopo il nome della funzione, **vanno racchiusi tra**

parentesi tonde e, se più di uno, vanno separati con il *separatore di elenco*, che in italiano è il punto e virgola.

Gli argomenti di una funzione possono essere:

- valori, cioè testi o numeri (a seconda della funzione);
- formule, che restituiscono valori;
- riferimenti a celle contenenti valori;
- altre funzioni.

La definizione degli argomenti di ciascuna funzione e cioè:

- il loro significato;
- il loro tipo, cioè se sono testi, numeri, riferimenti;
- l'ordine con cui appaiono, se più di uno; prende il nome di **sintassi** della funzione.

Quando un argomento di una funzione è a sua volta un'altra funzione, si dice che le funzioni sono **nidificate**.

La funzione più semplice è la funzione SOMMA. La sua sintassi è:

=SOMMA(arg1; arg2; ... argN),

dove arg1; arg2; ... argN sono da 1 a 30 argomenti che si vogliono sommare. Gli argomenti dal secondo in poi sono *facoltativi* (cioè possono essere omessi). Tutti gli argomenti possono essere numeri, riferimenti a celle (*compresi gli intervalli di celle*) o altre funzioni che restituiscano numeri.

Ad esempio sono funzioni valide:

=SOMMA(5;3;40)	somma tre argomenti che sono numeri
=SOMMA(A2;3;40)	somma un argomento riferimento e due argomenti numeri
=SOMMA(A2:C8)	somma un argomento riferimento che è un intervallo (si noti che esso è comunque <u>un singolo argomento</u>)
=SOMMA(C3:C20;3*2;4-A3)	somma un intervallo, una formula con numeri e una formula con riferimenti
=SOMMA(A1;C3:C20;SOMMA(D3;D20))	somma un riferimento, un intervallo e un'altra funzione (in questo caso non ha particolare senso sommare una somma, ma è per far vedere che le funzioni si possono nidificare)

IL FOGLIO DI ESEMPIO

Faccio un esempio del tutto classico, ma che si presta bene a vedere l'uso delle formule nel caso di *calcoli ripetitivi*. È infatti assai comune che, all'interno di un foglio di calcolo, le formule non siano "sparse" e ognuna per conto suo (è cioè ben difficile che davvero si vogliano fare le somme che ho scritto poco fa a scopo didattico); al contrario spesso le formule si ripetono in modo sistematico con lievi varianti "strutturate", come risulterà nel corso dell'esempio.

Immaginiamo di avere la popolazione di una provincia della Lombardia e la sua superficie:

	A	B	C	D	E	F
1	Censimenti	<i>Abitanti</i>				
2	Provincia	Censimento 1931	Censimento 1951	Censimento 1971	Censimento 1991	Superficie
3	Bergamo	612'891	696'626	829'019	932'104	2'759.3

Se voglio calcolare la densità, cioè gli abitanti al km², devo dividere la popolazione per la superficie. Ad esempio per il censimento del 1991, scriverò:

	J
1	<i>Densità</i>
2	ab/km ³ 1991
3	=E3/F3

In J3 ottengo il valore 338 (cioè 932'104/2'759.3)

Se poi ho la popolazione totale della Lombardia:

	A	B	C	D	E	F
1	Censimenti	<i>Abitanti</i>				
2	Provincia	Censimento 1931	Censimento 1951	Censimento 1971	Censimento 1991	Superficie
14	LOMBARDIA	5'595'915	6'566'154	8'543'387	8'856'074	23'855.5

e voglio sapere che percentuale degli abitanti lombardi vive in provincia di Bergamo (sempre nel 1991), scriverò:

	N
1	
2	1991
3	=E3/E14

E in N3 ottengo il 10.5% (una volta che ho formattato il campo come percentuale).

Fin qui ho fatto due calcoli fra loro indipendenti, e basta. Adesso però voglio considerare tutte e 11 le province della Lombardia (e tutti e 4 i censimenti) e fare gli stessi calcoli per tutte. È chiaro che i calcoli saranno simili, ma non uguali.

Ad esempio:

➤ Per la densità:

a pari provincia, la superficie sarà sempre la stessa, mentre la popolazione cambierà con il censimento.

➤ Per la ripartizione percentuale:

a pari censimento, la popolazione totale della Lombardia sarà sempre la stessa, mentre la popolazione cambierà con la provincia.

Comincio a considerare i miei dati:

	A	B	C	D	E	F
1	Censimenti	Abitanti				
2	Provincia	Censimento 1931	Censimento 1951	Censimento 1971	Censimento 1991	Superficie
3	Bergamo	612'891	696'626	829'019	932'104	2'759.3
4	Brescia	721'890	858'243	957'686	1'044'544	4'782.5
5	Como	318'758	361'667	476'209	522'147	1'288.1
6	Cremona	369'175	381'816	334'281	327'970	1'770.6
7	Lecco	176'002	200'837	244'254	273'536	778.8
8	Lodi	167'420	180'436	175'844	184'025	782.2
9	Mantova	403'422	424'753	376'892	369'630	2'338.8
10	Milano	1'807'367	2'324'717	3'727'841	3'738'685	1'980.1
11	Pavia	487'323	506'511	526'389	490'898	2'964.7
12	Sondrio	140'218	153'493	169'149	175'496	3'211.9
13	Varese	391'449	477'055	725'823	797'039	1'198.7

Tutti questi sono **valori** (non formule). Parte sono **testi** (sono allineati a sinistra) e parte **numeri** (sono allineati a destra). Sono in particolare testi:

➤ le "**intestazioni di colonna**" (che dicono di che censimento sto parlando o che sto considerando la superficie);

➤ il primo "**campo**" (sinonimo di colonna) che dice di che provincia sto parlando.

Ora calcolo le densità; scriverò:

	G	H	I	J
1	<i>Densità</i>			
2	ab/km ³ 1931	ab/km ³ 1951	ab/km ³ 1971	ab/km ³ 1991
3	=B3/\$F3	=C3/\$F3	=D3/\$F3	=E3/\$F3
4	=B4/\$F4	=C4/\$F4	=D4/\$F4	=E4/\$F4
5	=B5/\$F5	=C5/\$F5	=D5/\$F5	=E5/\$F5
6	=B6/\$F6	=C6/\$F6	=D6/\$F6	=E6/\$F6
7	=B7/\$F7	=C7/\$F7	=D7/\$F7	=E7/\$F7
8	=B8/\$F8	=C8/\$F8	=D8/\$F8	=E8/\$F8
9	=B9/\$F9	=C9/\$F9	=D9/\$F9	=E9/\$F9
10	=B10/\$F10	=C10/\$F10	=D10/\$F10	=E10/\$F10
11	=B11/\$F11	=C11/\$F11	=D11/\$F11	=E11/\$F11
12	=B12/\$F12	=C12/\$F12	=D12/\$F12	=E12/\$F12
13	=B13/\$F13	=C13/\$F13	=D13/\$F13	=E13/\$F13

Ottenendo i seguenti valori:

	G	H	I	J
1	<i>Densità</i>			
2	ab/km ³ 1931	ab/km ³ 1951	ab/km ³ 1971	ab/km ³ 1991
3	222	252	300	338
4	151	179	200	218
5	247	281	370	405
6	209	216	189	185
7	226	258	314	351
8	214	231	225	235
9	172	182	161	158
10	913	1'174	1'883	1'888
11	164	171	178	166
12	44	48	53	55
13	327	398	606	665

Si nota subito la presenza dei simboli \$ all'interno dei riferimenti. Occorre introdurre i riferimenti assoluti e quelli relativi.

RIFERIMENTI ASSOLUTI E RELATIVI

La differenza tra riferimenti assoluti e relativi si ha ***se e solo se le celle contenenti i riferimenti vengono copiate e incollate in un'altra posizione.***

I riferimenti assoluti non vengono modificati durante l'operazione di copia e incolla. I riferimenti relativi vengono invece modificati.

I riferimenti assoluti hanno i simboli di \$.

I riferimenti relativi non li hanno.

Un riferimento può essere relativo o assoluto *separatamente per la riga o la colonna*:

<i>rif.</i>	<i>colonna</i>	<i>riga</i>
A1	relativo	relativo
A\$1	relativo	assoluto
\$A1	assoluto	relativo
\$A\$1	assoluto	assoluto

Se si sta scrivendo o modificando una formula, si passa tra i vari tipi di riferimenti usando il tasto **F4** (più volte in successione fino ad ottenere la combinazione desiderata).

Un riferimento assoluto identifica sempre la medesima cella (rispettivamente la medesima riga o colonna, se è assoluto solo per la riga o la colonna).

Un riferimento relativo identifica la posizione della cella indicata, *come scarto rispetto alla cella in cui il riferimento è posto.*

Quindi se in A3 scrivo =B\$5, mi sto comunque riferendo alla cella della colonna B e della riga 5.

Se scrivo =B5 mi sto riferendo alla cella *1 colonna più a destra e 2 righe più sotto della cella A3* (perché da A a B c'è una colonna e da 3 a 5 ci sono 2 righe).

Se copio la cella A3 in D8, =B\$5 rimarrà immutato, mentre =B5 diventerà =E10.

Vediamo ora l'esempio della densità: la colonna della superficie non cambia al cambiare degli anni: il riferimento alla colonna F sarà assoluto. La colonna della popolazione cambia al cambiare degli anni: il riferimento alla colonna B, C, D, E sarà relativo. Al cambiare delle province, la riga cambia sia per la superficie sia per la popolazione: i riferimenti di riga saranno sempre relativi.

Questo significa che la densità di Bergamo di un certo anno è data dalla popolazione che sta sulla stessa riga, ma 5 colonne prima (ad es. per il 1991, da J ad E sono appunto cinque colonne), divisa per la superficie che sta sulla stessa riga, ma sempre sulla colonna F (indipendentemente dall'anno).

*Attenzione: questo significa che io devo scrivere a mano solo una prima formula: ad esempio quella in G3. Poi **la copierò e la incollerò** in tutte le altre celle, e i riferimenti saranno sempre corretti.*

In alternativa a copiare e incollare, posso trascinare il **"quadrantino di trascinamento"**, cioè quel piccolo quadratino che sta allo spigolo inferiore destro del bordo spesso di una cella quando essa è selezionata.

Per calcolare la ripartizione percentuale, comincio a sommare la popolazione:

	A	B	C	D	E
1	Censimenti	Abitanti			
	Provincia	Censimento 1931	Censimento 1951	Censimento 1971	Censimento 1991
13	LOMBARDIA	=SOMMA(B3:B13)	=SOMMA(C3:C13)	=SOMMA(D3:D13)	=SOMMA(E3:E13)

Ottingo:

	A	B	C	D	E
1	Censimenti	Abitanti			
2	Provincia	Censimento 1931	Censimento 1951	Censimento 1971	Censimento 1991
14	LOMBARDIA	5'595'915	6'566'154	8'543'387	8'856'074

E le formule sono quindi:

	K	L	M	N
1	Ripartizione percentuale			
2	1931	1951	1971	1991
3	=B3/B\$14	=C3/C\$14	=D3/D\$14	=E3/E\$14
4	=B4/B\$14	=C4/C\$14	=D4/D\$14	=E4/E\$14
5	=B5/B\$14	=C5/C\$14	=D5/D\$14	=E5/E\$14
6	=B6/B\$14	=C6/C\$14	=D6/D\$14	=E6/E\$14
7	=B7/B\$14	=C7/C\$14	=D7/D\$14	=E7/E\$14
8	=B8/B\$14	=C8/C\$14	=D8/D\$14	=E8/E\$14
9	=B9/B\$14	=C9/C\$14	=D9/D\$14	=E9/E\$14
10	=B10/B\$14	=C10/C\$14	=D10/D\$14	=E10/E\$14
11	=B11/B\$14	=C11/C\$14	=D11/D\$14	=E11/E\$14
12	=B12/B\$14	=C12/C\$14	=D12/D\$14	=E12/E\$14
13	=B13/B\$14	=C13/C\$14	=D13/D\$14	=E13/E\$14
14	=B14/B\$14	=C14/C\$14	=D14/D\$14	=E14/E\$14

Da cui ottengo:

	K	L	M	N
1	<i>Ripartiz %</i>			
2	1931	1951	1971	1991
3	11.0%	10.6%	9.7%	10.5%
4	12.9%	13.1%	11.2%	11.8%
5	5.7%	5.5%	5.6%	5.9%
6	6.6%	5.8%	3.9%	3.7%
7	3.1%	3.1%	2.9%	3.1%
8	3.0%	2.7%	2.1%	2.1%
9	7.2%	6.5%	4.4%	4.2%
10	32.3%	35.4%	43.6%	42.2%
11	8.7%	7.7%	6.2%	5.5%
12	2.5%	2.3%	2.0%	2.0%
13	7.0%	7.3%	8.5%	9.0%
14	100.0%	100.0%	100.0%	100.0%

In questo caso è assoluto il riferimento alla riga dove ho la popolazione della Lombardia (la riga è la stessa per tutte le province). Sono invece relativi i riferimenti alle colonne (la popolazione, sia delle province, sia della Lombardia, è sempre quella di 5 colonne più a sinistra). Sono relativi anche i riferimenti alle righe delle province, perché la ripartizione percentuale sta sempre sulla stessa riga della corrispondente popolazione (cioè sta 0 righe sotto, ragionando in termini di "scarto").

Si noti da ultimo che per la Lombardia la percentuale restituisce correttamente il 100%.

L'ESEMPIO DELL'ELENCO DI LIBRI

Un altro esempio classico è quello dell'elenco di libri. In esso cominceremo ad usare non solo delle semplici formule (cioè formule con le quattro operazioni), ma anche delle **funzioni**, un po' più complesse della funzione SOMMA vista in precedenza.

L'elenco di libri è organizzato in righe e colonne. Questa organizzazione è detta del tipo a **tabella** o "**foglio dati**": le righe contengono gli *oggetti fisici* di cui si sta parlando (qui i libri), le colonne contengono le *informazioni omogenee* che si vogliono dare su tali oggetti (qui l'autore, il titolo del libro, l'edizione, ecc.).

Le righe sono dette **record** e le colonne sono dette **campi**.

La prima riga contiene i nomi dei campi (detti anche "*intestazioni di colonna*").

I primi quattro campi contengono **testo**, i rimanenti tre contengono **numeri**. In particolare l'anno non ha nessuna formattazione (è già leggibile ... al naturale), la data è appunto formattata come data (ma è sempre un numero), il prezzo è formattato con il separatore delle migliaia.

	A	B	C	D	E	F	G
1	Autore	Titolo	Luogo	Edizione	Anno	Data	Prezzo
2	FOSTER, Alan Dean	Alien	Milano	Bompiani	1980	10/9/96	2'800
3	BANTI, Anna	Lavinia fuggita	Milano	La Tartaruga	1996	27/11/96	4'000
4	PINTER, Harold	La donna del tenente francese	Torino	Einaudi	1982	2/10/96	7'500
5	BERTONCELLI, Riccardo	Led Zeppelin	Firenze	Giunti	1995	10/12/96	8'000

AGGIUNTA DI CAMPI CALCOLATI

Immaginiamo che il prezzo dei libri sia dato IVA esclusa e noi vogliamo sapere il prezzo IVA compresa. Dal momento che ci interessa il prezzo *di ciascun libro*, dovremo scrivere una formula, nella prima colonna libera, *per ciascuna riga* (record) del nostro elenco. Questo significa in pratica che *stiamo inserendo un nuovo campo* (che è detto in particolare "**campo calcolato**", in quanto i suoi valori sono legati a quelli degli altri campi attraverso un'operazione matematica).

Se l'IVA dei libri è il 12%, il prezzo IVA compresa è dato da:

$$\text{prezzo intero} + 12\% * \text{prezzo intero}$$

o, se non vi spaventa, da:

prezzo intero * 1.12

Posso quindi scrivere (e comincerò a scriverlo in H2):

=G2+12%*G2

Non occorrono segni \$, e posso trascinare questa formula verso il basso; avrò automaticamente G3 in H3, G4 in H4, ecc.

Se però l'IVA è aumentata nel corso del tempo, ed era il 4% per i libri editi prima del 1995, la formula dipenderà da quanto c'è scritto nel campo "Anno":

=SE(E2<1995;G2+4%*G2;G2+12%*G2)

Ho introdotto una bellissima funzione, la funzione SE:

SE(test;se_vero;se_falso)

Se test è vero, la funzione SE restituisce se_vero; se test è falso, la funzione SE restituisce se_falso. È importante capire il significato concettuale dei tre argomenti (meglio: è importante capire il significato *degli argomenti*, in via del tutto generale).

Il primo argomento, test, è una qualsiasi espressione che possa restituire VERO o FALSO.

3>2 restituisce sempre VERO 3<2 restituisce sempre FALSO. Questo è *comunque corretto*, ma non è utilissimo!

Ma E2<1995 restituisce VERO o FALSO a seconda del contenuto della cella E2: questo è *molto* utile!

Se test ha dato VERO, viene letto l'argomento se_vero. Attenzione: se_vero è solo un nome, una descrizione dell'argomento, fatta per ricordarmene il significato: l'argomento è in questo caso una qualunque espressione che restituisca un valore, numerico o di testo. Se E2 è minore di 1995, in H2 (dove ho scritto la formula) è come se avessi scritto solo =G2+4%*G2.

Se test ha dato FALSO, viene letto l'argomento se_falso. Se E2 è maggiore o uguale a 1995, in H2 è come se avessi scritto solo =G2+12%*G2.

Così com'è, la formula usa un'IVA del 4% fino al 1994 e del 12 dal 1995. Se volessi mettere il 1995 "dall'altra parte", basterebbe usare un minore o uguale anziché un solo minore:

=SE(E2<=1995;G2+4%*G2;G2+12%*G2)

Cioè, più in generale:

uguale	=
minore	<
minore o uguale	<=
maggiore	>
maggiore o uguale	>=
diverso	<>

Ripeto: test è un argomento che deve restituire vero o falso (3+5 non può essere un test!), se_vero e se_falso possono restituire non solo numeri ma anche testo.

Come funzionerà dunque la formula che segue?

=SE(G2>30000;"Che libro costoso!";"Spendo poco")

(notare che un testo da riportare tal quale va scritto tra virgolette: in termini tecnici esso si chiama **stringa**)

Se ho ricopiato la formula (quella ... seria) per tutte le righe, può essere spiacevole accorgermi che ho sbagliato aliquota IVA (oppure anno). Più in generale, noto che nel foglio non è scritto da nessuna parte in maniera chiara quali aliquote ho considerato e per che anni.

Prendo un altro foglio della cartella di lavoro (se non ce ne sono, basta che scelga *Inserisci, Foglio di lavoro*) e compilo un prospetto del tipo:

	A	B	C
1			
2	Anni fino a:	Quota IVA	
3	1995	4%	
4		12%	

(notare che ho scritto l'anno da solo, e il "fino a" l'ho messo a parte, nell'intestazione della colonna: se avessi scritto "fino al 1995" nella stessa cella A3, avrei mescolato informazioni disomogenee – testo e numeri – e non avrei potuto usare quella cella per un conto)

Il foglio si chiama, ad esempio, Parametri (per rinominarlo basta fare doppio clic sulla sua "linguetta") La mia funzione SE diventa:

=SE(E2<Parametri!\$A\$3;G2+G2*Parametri!\$B\$3;G2+G2*Parametri!\$B\$4)

Sembra più complicata, ma è molto più versatile. Basta cambiare o anno o quote nel prospetto di Parametri, che tutto il campo Prezzo IVA compresa verrà automaticamente aggiornato.

Notare (fondamentale!) il dollaro per tutti e tre i riferimenti di Parametri: essi devono infatti rimanere immutati per ogni prezzo.

Come aggiunta, notare anche la sintassi per indicare il **riferimento ad una cella di un altro foglio**:

NomeFoglio!A1

cioè un punto esclamativo **Errore. Il segnalibro non è definito.** che separa il nome del foglio dal normale riferimento (con o senza dollaro).

Se il foglio contiene degli spazi, va racchiuso tra virgolette semplici (cioè apostrofi): **Errore. Il segnalibro non è definito.**

'Nome foglio'!A1

Se poi il riferimento è a un foglio di un'altra cartella (cioè un altro file) **Errore. Il segnalibro non è definito.:**

[NOMEFILE.XLS]NomeFoglio!A1 oppure '[NOMEFILE.XLS]Nome Foglio'!A1

Se ottengo i riferimenti con il mouse (facendo clic sulle celle corrispondenti), vengono di sicuro scritti giusti!

E se le aliquote IVA sono invece tre:

	A	B	C
1			
2	Anni fino a:	Quota IVA	
3	1995	4%	
4	1997	12%	
5		20%	

Scrivendo addirittura la versione riferita al prospetto, avrò:

=SE(E2<Parametri!\$A\$3;G2+G2*Parametri!\$B\$3;SE(E2<Parametri!\$A\$3;G2+G2*Parametri!\$B\$4;G2+G2*Parametri!\$B\$5))

O, nella forma più breve:

=SE(E2<1995;G2+G2*4%;SE(E2<1997;G2+G2*12%;G2+G2*20%))

Cioè, sintatticamente:

=SE(test1;se_vero1;SE(test2;se_vero2;se_falso2))

La parte evidenziata costituisce l'argomento se_falso della prima funzione SE, ma è essa stessa una funzione SE. Due funzioni (non necessariamente uguali) una dentro nell'altra si dicono **nidificate**. È possibile nidificare anche *più di due* funzioni.

Non esiste mai un solo modo per ottenere un risultato: visto che G2+G2* compare più volte, posso scrivere anche:

=G2+G2*SE(E2<1995;4%;SE(E2<1997;12%;20%))

In questo caso i se_vero e se_falso non sono più espressioni, ma semplici numeri. In compenso, è la funzione SE che non è più da sola ma fa parte di un'espressione (la moltiplica per G2).

ESTRAZIONE DI INFORMAZIONI DI RIEPILOGO

Un'operazione spesso duale a quella di aggiungere un campo calcolato è quella di introdurre uno specchietto di riepilogo. In questo caso le formule non verranno scritte all'interno della tabella, ma da un'altra parte (anche su un altro foglio).

Per esempio posso voler sapere numero e costo di tutti i libri di una certa casa editrice, per esempio la Mondadori. Scriverò:

=CONTA.SE(D2:D163;"Mondadori")

sapendo che la sintassi della funzione CONTA.SE è:

=CONTA.SE(intervallo;criteri)

Intervallo è l'intervallo di celle che devo contare e criteri è un'espressione che devo cercare in ciascuna cella dell'intervallo. Notare che criteri, a differenza di test, **non** restituisce vero o falso. In questo caso infatti il test è implicito nella funzione: esso sarebbe del tipo *ciascuna_cella=criteri*. In questo modo la funzione CONTA.SE viene semplificata; il suo limite è tuttavia che il test viene comunque fatto solo con l'uguale (non con maggiore, minore, diverso).

Se sono sicuro che la colonna D non contiene altre informazioni "di troppo", una formula:

=CONTA.SE(D:D;"Mondadori")

ha il vantaggio che non è da aggiustare se aggiungo un libro sulla riga 164.

In più i segni \$ qui non sono indispensabili, ma non guasterebbero comunque.

E per sapere qual è il costo totale dei libri della Mondadori:

=SOMMA.SE(D2:D163;"Mondadori";G2:G163)

oppure:

=SOMMA.SE(D:D;"Mondadori";G:G)

Dove la funzione SOMMA.SE è molto simile alla precedente, solo che invece di contare tutte le volte che viene trovato che *ciascuna_cella=criterio*, si somma quello che c'è scritto in *int_somma*:

=SOMMA.SE(intervallo;criterio;int_somma)

Ma se voglio fare una cosa che vada bene non solo per la Mondadori (una cosa che, come si dice, sia "parametrica"), posso farmi un nuovo prospetto, del tipo:

	L	M	N
1			
2	Casa editrice	Numero libri	Costo totale
3	Mondadori	=CONTA.SE(D:D;L3)	=SOMMA.SE(D:D;L3;G:G)
4	Einaudi	=CONTA.SE(D:D;L4)	=SOMMA.SE(D:D;L4;G:G)
5	Rizzoli	=CONTA.SE(D:D;L5)	=SOMMA.SE(D:D;L5;G:G)

L'elenco può crescere a piacere: le formule CONTA.SE e SOMMA.SE possono essere trascinate verso il basso fin dove desiderato.

Un altro caso è quello in cui ho delle informazioni semplicemente date da crocette (o da qualunque altro segno all'interno di una cella):

	A	B	C
1	Autore	Titolo	Disponibile
2	FOSTER, Alan Dean	Alien	X
3	BANTI, Anna	Lavinia fuggita	
4	PINTER, Harold	La donna del tenente francese	X
5	BERTONCELLI, Riccardo	Led Zeppelin	X

Se voglio sapere quanti sono i libri disponibili, scrivo in una cella:

=CONTA.VALORI(C2:C5)

dal momento che la sintassi è:

=CONTA.VALORI(val1;val2;...)

dove, proprio come nella funzione SOMMA, val1, val2,... sono qualunque cosa (per esempio degli intervalli) che possa essere contata.

Attenzione, però: se preferisco usare il riferimento a tutta la colonna (che non mi dà problemi se aggiungo un libro in fondo), devo scrivere:

=CONTA.VALORI(C:C)-1

dove il -1 mi sottrae il valore "Disponibile" (cioè l'intestazione del campo), che altrimenti verrebbe contato anch'esso.

FUNZIONALITÀ DI EXCEL

SPOSTAMENTO TRA LE CELLE

Oltre a potersi muovere normalmente con i tasti cursore, le barre di scorrimento e il mouse, si segnala che:

- facendo doppio clic su un bordo di una *cella non vuota* (quando cioè il puntatore ha forma di freccia e *non* di grosso +) ci si sposta all'ultima cella non vuota dalla parte del bordo su cui si è fatto doppio clic. Ad esempio, facendo doppio clic sul bordo inferiore di una cella, ci si porta all'ultima cella non vuota sotto tale cella:
- facendo doppio clic su un bordo di una *cella vuota* ci si sposta all'ultima cella vuota dalla parte del bordo su cui si è fatto doppio clic.

SELEZIONE DI CELLE

La selezione di intervalli di celle si effettua in molti modi: si richiamano qui i casi più tipici.

- Clic sulla prima cella e Shift + Clic sull'ultima: si seleziona l'intero intervallo dalla prima all'ultima;
- Clic sulla prima cella e Ctrl + Clic su altre celle: si selezionano tali celle *anche non contigue* (selezione "discontinua");
- Clic + Trascina sul primo intervallo e Ctrl + Trascina su altri intervalli: si selezionano intervalli non contigui (analogamente al caso precedente);
- Clic sulla prima cella e Shift + doppio clic su un bordo della stessa cella: si seleziona l'intero intervallo da tale cella all'ultima piena (o vuota, se la prima era vuota) dalla parte in cui si è fatto doppio clic (analogamente a quanto visto al paragrafo precedente);
- Clic sulla prima cella e Shift + tasti cursori: si estende la selezione (usando la tastiera anziché il mouse).

SPOSTAMENTO E COPIA DI CELLE

In Excel è molto semplice spostare o copiare celle all'interno di fogli di lavoro. Occorrono tuttavia alcune precisazioni su quello che accade ai riferimenti una volta spostate o copiate le celle.


Copia elementare delle celle

Si procede in due passaggi:

1. Si copia la cella o l'intervallo di origine:

Menu: Modifica, Copia


Scelta rapida: Ctrl + C

Pulsante: 

2. Si incolla la cella o l'intervallo copiato sulla destinazione:

Menu: Modifica, Incolla

Scelta rapida: Ctrl + V

Pulsante: 

Spostamento elementare delle celle

È come il caso della copia, ma il primo passaggio usa la funzione di "taglia":

1. Si taglia la cella o l'intervallo di origine:

Menu: Modifica, Taglia

Scelta rapida: Ctrl + X

Pulsante: 

2. Si incolla la cella o l'intervallo copiato sulla destinazione (come caso precedente).

Se si copia (o taglia) un intervallo, non è necessario selezionare un intervallo uguale come destinazione: basta selezionare la prima cella in altro a sinistra, e la selezione verrà automaticamente espansa in misura pari alle celle da incollare. Occorre fare attenzione a che questo non sovrascriva celle che si vuole mantenere.

È possibile copiare anche intere righe o colonne, selezionando rispettivamente l'intestazione di riga (numero) o di colonna (lettera) corrispondente.

Riferimenti per le celle copiate

Si immagini la struttura seguente:

	A	B	C
1			
2	1	=A2	
3	2		
4			

Se taglio la cella A2 e la incollo (ad esempio) in A4, la cella B2 continuerà a restituire 1 e conterrà la formula =A4. In altre parole, **tagliando** e incollando delle celle, *le celle spostate si "portano dietro" i riferimenti*. Le altre celle continueranno cioè a far riferimento alle stesse "informazioni" (il numero 1, in questo caso), indipendentemente dal fatto che queste informazioni si sono spostate nel foglio.

Questo vale indipendentemente dal fatto che il riferimento =A2 sia assoluto o relativo (riferimenti assoluti e relativi entrano in gioco se copio le celle che li contengono – in questo caso B2 – e *non* se tocco le celle a cui si riferiscono).

Se copio la cella A2 e la incollo in A4, la cella B2 continuerà a restituire 1 e conterrà la formula =A2. In altre parole, **copiando** e incollando delle celle, *le altre celle che fanno riferimento alle celle copiate non vengono aggiornate*.

Se ci si pensa, la cosa è ovvia: se taglio una cella, quella non esiste più nella posizione di origine, ed è *necessario* che il riferimento diventi alla posizione di destinazione. Se la copio, essa continua ad esistere nella posizione di origine, e si può continuare a riferirsi ad essa.

Un altro caso: ora *taglio la cella A3 e la incollo sopra la cella A2*. La cella A2 "vecchia" (quella con l'1) non esiste più: la cella B2 che si riferiva ad essa restituirà quindi un valore di errore di riferimento: **#RIF!**.

Se invece *copio la cella A3 e la incollo sopra la cella A2*, la cella B2 continua a contenere la formula =A2 e restituisce di conseguenza 2, cioè il nuovo valore della cella A2 stessa.

Un caso pratico in cui bisogna tenere conto di ciò è il seguente: immaginiamo di avere un elenco di prodotti e relativi prezzi. Alcune formule fanno riferimento a quei prezzi (ad esempio li sommano). Se devo aggiornare i prezzi, prelevandoli da un altro foglio, devo necessariamente *copiarli (e non tagliarli!)*, altrimenti tutte le formule restituiranno un errore #RIF!.

Si noti come tutti i casi valgano anche quando si copia o sposta tra fogli diversi della stessa cartella di lavoro.

Se poi si spostano (*non*: copiano) delle celle in un'altra cartella di lavoro, cioè in un altro file, occorre sapere che le celle che si riferivano a quelle spostate ora appariranno **"collegate"** alle celle di quest'altro file. Riaprendo il primo (con quest'ultimo chiuso) verrà chiesto se "si desidera aggiornare i riferimenti a file non aperti" (cioè controllare che i valori mostrati nel primo file siano aggiornati rispetto ad eventuali modifiche apportate nel frattempo al secondo). Questo discorso vale anche nel caso si creino volontariamente dei riferimenti ad un altro file, non come effetto di un'operazione di taglia e incolla. I file collegati al file corrente sono elencati nella finestra che appare scegliendo *Modifica, Collegamenti*.

Spostamento e copia con il mouse

Con il mouse le operazioni di spostamento o copia si effettuano semplicemente trascinando il bordo della cella (o della selezione, se questa è un intervallo), quando il puntatore ha forma di freccia (*non* di grosso +).

Si hanno le seguenti opzioni:


- **Trascina e basta** corrisponde a **tagliare** e incollare. Dal momento che si sta tagliando, valgono tutte le cose dette sopra (le celle spostate si "portano dietro" i riferimenti e fanno sì che le celle sovrascritte non esistano più);
- **Ctrl + Trascina** corrisponde a **copiare** e incollare. In genere questa operazione è meno rischiosa della precedente, in tutti i casi in cui la destinazione non era vuota, ma conteneva già delle informazioni;
- **Shift + Trascina** corrisponde a tagliare e **inserire le celle tagliate** nella posizione evidenziata dal tratteggio a doppio T che appare trascinando. In questo caso non si sovrascrive alcuna cella, ma "si fa spazio" per le celle che si stanno spostando, semplicemente facendo scorrere le altre verso il basso o verso destra (a seconda se la doppia T è orizzontale o verticale).

Si noti come questo sia il sistema migliore per riordinare delle colonne (o delle righe): basta selezionare la colonna (o la riga) e portarla nella nuova posizione. Se non si tenesse premuto Shift, la colonna ne sovrascriverebbe un'altra (e lascerebbe un "buco" dove era prima).

- **Ctrl + Shift + Trascina** corrisponde a copiare e **inserire le celle copiate** nella posizione evidenziata dal tratteggio a doppio T. È come il caso precedente, ma le celle vengono lasciate anche nella posizione di origine. Questo è un metodo valido per duplicare delle colonne (o delle righe), senza correre il rischio di sovrascriverne altre.

"Incolla speciale" per incollare solo alcuni aspetti di una cella

Se dopo aver copiato una cella (o un intervallo), anziché scegliere *Incolla* (o il corrispondente pulsante o Ctrl + V), si sceglie *Modifica, Incolla speciale*, si ottengono le opzioni della seguente finestra di dialogo, utili in molti casi:



Le opzioni più tipicamente vantaggiose sono le seguenti:

- **Incolla Valori**: vengono incollati i valori delle celle. Celle che contengono numeri o testi vengono incollate normalmente, ma, se una cella contiene una formula, ne viene incollato **il solo risultato**, cioè appunto il "valore".

Ad esempio ho il caso seguente:

	A	B	C
1		Prezzo	
2		5'000	
3		10'000	
4		15'000	
5	Totale:	=SOMMA(\$B\$2:\$B\$4)	

Devo ricopiare il totale in un foglio di riepilogo: se copiassi e incollassi la cella B5 nell'altro foglio, essa farebbe ora riferimento alle celle B2:B4 di quest'altro foglio, restituendomi un valore errato (se non ci fossero i dollari, farebbe addirittura riferimento non a B2:B4, ma ad un intervallo da una a tre righe sopra quella dove incollo!)

Se incollo il solo risultato ho risolto il problema. Naturalmente non ho più l'aggiornamento automatico (se cambio B2, il risultato copiato e incollato come valore *non* si aggiorna!), ma in alcuni casi questo può essere ininfluenza, o addirittura proprio ciò che voglio.

➤ **Incolla Formati:** viene incollato tutto il formato della cella copiata (grassetto, corsivo, carattere, bordi, colori, formato numerico, ecc.). Questo è spesso assai utile per dare a due celle (o due intervalli) un aspetto omogeneo.

Esiste anche un pulsante "**copia formato**" che funziona come quello di Word: si seleziona l'intervallo da cui copiare, poi si preme il pulsante, e infine si seleziona la destinazione.

Pulsante "**copia formato**": 

Si noti che, in entrambi i casi, è possibile applicare più formati contemporaneamente: se seleziono ad esempio come origine due celle, la prima in grassetto e la seconda in corsivo, e poi incollo il formato su un intervallo rettangolare largo due celle e alto a piacere, tutte le celle della prima colonna saranno in grassetto, tutte quelle della seconda in corsivo.

➤ **Operazioni:** con questa opzione è facile effettuare degli **aggiornamenti** a gruppi di celle. Si segua in proposito l'esempio.

Ho la struttura che segue:

	A	B	C
1	Prezzo		1.2
2	5'000		
3	10'000		
4	15'000		
5			

Devo aggiornare la colonna dei prezzi, applicando un aumento del 20% (ciò equivale a moltiplicare i prezzi per 1.2). Scrivo 1.2 in una cella qualunque, ad esempio la C1. Copio C1. Seleziono A2:A4 e scelgo *Incolla speciale* con operazione *Moltiplica*. Ottengo quanto desiderato:

	A	B	C
1	Prezzo		1.2
2	6'000		
3	12'000		
4	18'000		
5			

Il valore 1.2 a questo punto può essere cancellato. La stessa cosa funziona naturalmente se devo sommare o sottrarre un valore costante ad un intervallo di celle o se devo dividere i valori nell'intervallo per un certo fattore. Si noti che i valori che voglio modificare sono *la destinazione* su cui incollo. Il valore che copio è invece quello per cui voglio moltiplicare, dividere, ecc. e lo scrivo in una cella qualsiasi.

In casi più complessi potrei voler moltiplicare, sommare ecc. intervalli delle stesse dimensioni (ad esempio due colonne).

Trasponi: con questa opzione posso scambiare fra loro le righe e le colonne. Se copio l'intervallo A1:A4 dell'esempio precedente, seleziono poi (per esempio) la cella D1 e scelgo *Incollo speciale* con l'opzione *Trasponi*, ottengo:

	D	E	F	G
1	Prezzo	6'000	12'000	18'000
2				

L'operazione funziona anche nel caso che l'intervallo non sia una colonna sola (o una riga sola) ma un'area rettangolare (cioè una **"matrice"**). Anche in essa verranno scambiate le righe con le colonne.

SUBTOTALI IN UNA TABELLA

In molti casi la semplice funzione SOMMA può risultare inadeguata, in quanto gli intervalli da sommare variano nel corso del foglio (ed eventualmente nel tempo) e occorrerebbe pertanto aggiustare a mano ogni argomento delle singole funzioni SOMMA. Si immagini ad esempio il caso in figura:

	A	B	C	D	E	F
1	Nazione	Città	Prezzo	Caratteristica	Data	Quantità
2	Austria	Salzburg	3'600	Così così	22/12/97	12
3	Austria	Wien	3'300	Così così	17/12/97	10
4	Francia	Lyon	2'800	Brutto	09/12/97	5
5	Francia	Marseille	550	Altro	01/05/97	12
6	Francia	Paris	840	Bello	01/12/97	10
7	Francia	Paris	1'200	Bello	05/12/97	5
8	Germania	Hamburg	1'800	Brutto	05/12/97	10
9	Germania	Hamburg	3'000	Così così	09/12/97	5
10	Germania	Hamburg	3'000	Così così	13/12/97	10
11	Germania	Koblenz	300	Altro	01/05/97	10
12	Germania	Mainz	1'000	Bello	01/12/97	6
13	Italia	Genova	700	Altro	01/05/97	14
14	Italia	Milano	1'000	Bello	01/12/97	8
15	Italia	Milano	1'500	Bello	05/12/97	10
16	Italia	Torino	2'500	Brutto	06/12/97	6
17	Svizzera	Basel	1'200	Bello	01/12/97	5
18	Svizzera	Lugano	2'100	Brutto	05/12/97	8
19	Svizzera	Luzern	250	Altro	30/04/97	5

Si desidera sapere il totale di prezzo e quantità *per ciascuna nazione*: naturalmente esistono più modi per ottenere tale informazione. Ad esempio si potrebbero usare funzioni del tipo =SOMMA.SE(A:A;"Austria";C:C), ma bisognerebbe applicare a mano il secondo argomento ("Austria", "Francia", ecc.). Vediamo ora come un risultato molto valido e di facile lettura possa essere ottenuto mediante la **"struttura"** e i **subtotali**.

ORDINAMENTO DI RECORD


Prima di procedere, si deve innanzitutto *ordinare i record* secondo la Nazione (nella figura precedente ciò è già stato fatto). L'ordinamento dei record è un'operazione molto comune (indipendentemente da che si voglia proseguire con il calcolo dei subtotali). Occorre selezionare l'area da ordinare e scegliere *Dati, Ordina*.

L'unica precauzione da seguire è quella di controllare sempre di *aver selezionato l'intera area che si vuole ordinare*. In alcuni casi Excel interpreta giustamente l'area da ordinare (ad esempio se non è selezionata alcuna cella, viene espansa la selezione fino a comprendere un rettangolo

di celle non vuote), ma occorre ricordare che un elenco che si è “spaiato” in seguito ad un ordinamento fatto solo su un suo pezzo diventa del tutto inutilizzabile!
In molti casi la cosa migliore è selezionare sempre *intere righe*.

CALCOLO DEI SUBTOTALI

A questo punto occorre scegliere *Subtotali* dal menu *Dati* e selezionare le opzioni come in figura:



Il risultato è mostrato nella figura seguente. Le righe di sottotale sono state inserite automaticamente. Anche la scritta “Totale Austria” e le analoghe successive sono state create automaticamente e inserite come testo. I subtotali (ad esempio 6'900 e 22 per l’Austria) compaiono, ovviamente, solo per i campi numerici (cioè rispettivamente Prezzo e Quantità).

	A	B	C	D	E	F
1	Nazione	Città	Prezzo	Caratteristica	Data	Quantità
2	Austria	Salzburg	3'600	Così così	22/12/97	12
3	Austria	Wien	3'300	Così così	17/12/97	10
4	Totale Austria		6'900			22
5	Francia	Lyon	2'800	Brutto	09/12/97	5
6	Francia	Marseille	550	Altro	01/05/97	12
7	Francia	Paris	840	Bello	01/12/97	10
8	Francia	Paris	1'200	Bello	05/12/97	5
9	Totale Francia		5'390			32
10	Germania	Hamburg	1'800	Brutto	05/12/97	10
11	Germania	Hamburg	3'000	Così così	09/12/97	5
12	Germania	Hamburg	3'000	Così così	13/12/97	10
13	Germania	Koblenz	300	Altro	01/05/97	10
14	Germania	Mainz	1'000	Bello	01/12/97	6
15	Totale Germania		9'100			41
16	Italia	Genova	700	Altro	01/05/97	14
17	Italia	Milano	1'000	Bello	01/12/97	8
18	Italia	Milano	1'500	Bello	05/12/97	10
19	Italia	Torino	2'500	Brutto	06/12/97	6
20	Totale Italia		5'700			38
21	Svizzera	Basel	1'200	Bello	01/12/97	5

I simboli grafici che compaiono alla sinistra dei numeri di riga indicano i **livelli** della **struttura**, che è stata creata parallelamente ai subtotali. Il segno MENO indica che la struttura è *espansa* (cioè mostrata tutta), il segno PIÙ (vedi figura successiva) indica invece che quel livello della struttura è stato *compresso* (cioè nascosto).

Per come è fatto l'elenco, esistono tre livelli:

- le singole città (livello 3)
- le nazioni (livello 2)
- il totale generale (livello 1)

I livelli vengono espansi o compressi facendo clic sul loro numero. Comprimendo il livello 3, ad esempio, si ha il risultato della figura seguente, in cui si vedono solo le nazioni:

1 2 3	A	B	C	D	E	F	
	1	Nazione	Città	Prezzo	Caratteristica	Data	Quantità
+	4	Totale Austria		6'900			22
+	9	Totale Francia		5'390			32
+	15	Totale Germania		9'100			41
+	20	Totale Italia		5'700			38
+	24	Totale Svizzera		3'550			18
-	25	Totale complessivo		30'640			151

Si noti come le righe 5:8, 10:14, ecc. appaiano *nascoste*.

È possibile espandere o comprimere singoli tratti della struttura, facendo clic sui simboli di PIÙ e MENO. Nella figura che segue è stata espansa la voce dell'Italia, ma non quella della Germania.

1 2 3	A	B	C	D	
	1	Nazione	Città	Prezzo	Caratteristica
+	4	Totale Austria		=SUBTOTALE(9;C2:C3)	
+	9	Totale Francia		=SUBTOTALE(9;C5:C8)	
+	15	Totale Germania		=SUBTOTALE(9;C10:C14)	
-	16	Italia	Genova	700	Altro
-	17	Italia	Milano	1000	Bello
-	18	Italia	Milano	1500	Bello
-	19	Italia	Torino	2500	Brutto
-	20	Totale Italia		=SUBTOTALE(9;C16:C19)	
-	21	Svizzera	Basel	1200	Bello
-	22	Svizzera	Lugano	2100	Brutto
-	23	Svizzera	Luzern	250	Altro
-	24	Totale Svizzera		=SUBTOTALE(9;C21:C23)	
-	25	Totale complessivo		=SUBTOTALE(9;C2:C23)	

Si noti inoltre la sintassi della funzione SUBTOTALE:

=SUBTOTALE(num_funzione;rif1;...)

dove num_funzione è un numero che indica che tipo di funzione si vuole applicare: 9 indica la *somma*, che è il caso più comune, ma dalla finestra "Subtotali" (prima figura di questo paragrafo) è possibile scegliere anche la *media*, il *conteggio*, ecc. Il conteggio tra l'altro può essere fatto anche su valori non numerici.

Rif1,... sono gli intervalli su cui agisce la funzione (somma, conteggio, ecc).

Se si inseriscono nuovi dati, si cambia campo per cui aggregare (la Nazione, nel nostro caso) o succedono comunque effetti strani, il sistema più veloce per ripristinare i subtotali è quello di eliminarli (scegliendo “Rimuovi tutti” dalla finestra *Dati, Subtotali*, e poi reinserendoli. Se sono stati aggiunti dati, ci si deve ricordare di *riordinare l’elenco*, prima di reinserire i subtotali.

FORMATTAZIONE DEI SUBTOTALI E STILI

Per formattare uniformemente i subtotali è possibile selezionare l’intero elenco, scegliere *Dati, Raggruppa e struttura, Impostazioni*, e quindi il pulsante “Crea”.

In questo modo ai subtotali vengono assegnati automaticamente gli “stili” *LivelloRiga_1, LivelloRiga_2*, ecc.

Gli **stili** sono un insieme di elementi di formattazione (ad esempio il grassetto, il corsivo, ecc.) che vengono applicati contemporaneamente. Anche la *percentuale* o le *migliaia* (cioè 2000 scritto come 2'000) sono stili.

Se si sono definiti degli stili, è facile attribuire al documento un aspetto omogeneo, formattando celle di contenuto simile con lo stesso stile.

Un ulteriore vantaggio degli stili è la possibilità di ridefinirli in base ad un esempio. Si segua questa procedura.

Una volta applicato lo stile, come descritto poco fa, si selezioni una cella contenente un subtotale (ad esempio quella del “Totale Austria”). Essa apparirà dello stile *LivelloRiga_2* (*). A questo punto si modifichi quella cella, cambiando dimensione del carattere, o colore o qualunque altra cosa. Si scelga quindi nuovamente lo stile *LivelloRiga_2* dalla casella degli stili.

Apparirà il messaggio:

“Ridefinire *LivelloRiga_2* in base alla selezione?”

Se si sceglie Sì, tutte le celle *LivelloRiga_2* (cioè tutti i totali delle altre nazioni) appariranno scritti con il nuovo carattere o colore.

Si lascia al lettore il compito di scoprire in quanti casi ciò può essere estremamente utile (non solo per i subtotali e non solo in Excel, detto per inciso, ma anche – e soprattutto! – in Word).

(*) Se la casella dello stile non è già presente, è possibile aggiungerla alla propria barra degli strumenti, scegliendo *Visualizza, Barre degli strumenti, Personalizza*, scegliendo quindi la categoria (*Formattazione testo*, in questo caso) e trascinando la casella in una posizione della barra degli strumenti (per es. di fianco alla casella del tipo di carattere).

I GRAFICI

I grafici permettono di rappresentare visivamente delle informazioni, agevolandone la lettura. Un grafico viene creato usando il **"Menu inserisci"**.


Occorre trascinare un rettangolo, che sarà quello contenente il grafico (non occorre preoccuparsi delle dimensioni, perché potrà essere ridimensionato in seguito a piacere).

Si immagini di voler tracciare il grafico dei seguenti dati:

	A	B
1	Regione	1931
2	Piemonte	3'457'731
3	Lombardia	5'595'915
4	Liguria	1'422'596
5	Veneto	3'487'109

Si è dunque selezionato l'intervallo A1:B5 e premuto il pulsante. Si è confermato l'intervallo di celle e scelto il grafico predefinito (istogramma normale).

Al passaggio 4 appare la seguente finestra di dialogo:




Notare che qui ci sono da scegliere tre opzioni importanti: Excel prova ad interpretare i dati e a fare una proposta, ma non è detto che sia quella giusta. In particolare:

- la serie di dati è in colonne (va bene la proposta di Excel, cfr. sotto per capire cosa è una serie di dati in righe);
- la prima colonna contiene le etichette dell'asse X: anche questo va bene, significa che in A:A non ci sono dati numerici (di cui fare il grafico), ma dei testi ("Piemonte", Lombardia", ...) che spiegano *a che cosa si riferirà* ciascun istogramma;

- la prima riga contiene il testo della legenda: qui Excel aveva sbagliato: aveva proposto 0 righe, tratto in inganno dal fatto che 1931 è un numero. È vero che è un numero, ma per noi è comunque un *elemento descrittivo*. In altre parole, Excel aveva pensato che "Regione" avesse 1931 abitanti; noi gli abbiamo detto che 1931 è qualcosa che dà informazioni sugli abitanti, e che questi – cioè quello che vogliamo disegnare nel grafico – cominciano da "Piemonte" (*ovvero dalla seconda riga*).

Ed ecco il grafico finale (con qualche piccolo aggiustamento: per modificare un grafico esistente, basta fare doppio clic su di esso per attivarlo, e quindi doppio clic sui singoli elementi: assi, griglie, istogrammi, ecc.):



In particolare, se selezioniamo una qualsiasi delle barre, vediamo la "formula" corrispondente, che assomiglia a tutti gli effetti ad una normale formula di Excel:

=SERIE(Grafici!\$B\$1;Grafici!\$A\$2:\$A\$5;Grafici!\$B\$2:\$B\$5;1)

Tenendo conto che il foglio che contiene i dati si chiama "Grafici", la sintassi è infatti:

=SERIE(Cella_della_legenda;Celle_delle_etichette_X;Celle_dei_Valori;Numero_della_serie)

dove dunque:

- B1 contiene la legenda della serie selezionata: "1931"
- A2:A5 contiene le etichette "Piemonte", "Lombardia", ...
- B2:B5 contiene i valori, cioè gli abitanti
- infine l'ultimo argomento indica il numero progressivo della serie (qui ce n'è una sola)


Si noti che, del rettangolo che avevamo selezionato all'inizio, manca la cella A1 (quella con scritto "Regioni"), che sarebbe la legenda delle etichette, e che nel grafico infatti non appare (in verità compare come testo dell'asse X, ma l'ho dovuta scrivere a mano, all'ultimo passaggio dell'autocomposizione).

Vediamo ora il caso in cui guardo tre censimenti insieme: 1931, 1961 e 1991:

	A	B	C	D
1	Regione	1931	1961	1991
2	Piemonte	3'457'731	3'914'250	4'290'412
3	Lombardia	5'595'915	7'406'152	8'831'264
4	Liguria	1'422'596	1'735'349	1'668'078
5	Veneto	3'487'109	3'846'562	4'363'157


Seleziono A1:D5, premo il pulsante dell'autocomposizione, trascino il rettangolo, arrivo ancora al quarto passaggio. Anche questa volta Excel è tratto in inganno dal fatto che gli anni sono dei numeri: devo dirgli io che la prima riga è il testo della legenda.

Qui però posso vedere il significato delle serie in riga o in colonna. Se, come prima, *scelgo le serie in colonna*, ottengo il grafico:




Questo grafico mi fa vedere chiaramente, regione per regione, quanto sono cresciuti gli abitanti nei 60 anni considerati (e vedo sufficientemente bene anche quanti sono gli abitanti di una regione rispetto all'altra: la Lombardia è cresciuta di più di tutti (ed è sempre stata la più popolosa). La Liguria, la meno popolosa, ha invece perso abitanti negli ultimi trent'anni.

Se invece *scelgo le serie in riga*, ecco che la legenda non contiene più gli anni, ma le regioni: ora leggo un po' più chiaramente, anno per anno, quale è il rapporto reciproco tra le quattro regioni.




Se quello che mi interessa è appunto il confronto tra le regioni (leggermente a scapito del loro andamento nel tempo, che vedevo bene nel primo grafico) posso addirittura "impilare" gli istogrammi (grafico sotto a sinistra, ottenuto scegliendo il sottotipo numero 3, anziché quello predefinito, che è il 6, al terzo passaggio dell'autocomposizione).

Una variante a volte vantaggiosa è il sottotipo 5 (grafico a destra) in cui gli istogrammi non sono solo impilati, ma anche tutti "risalati" a 100 (tanto è vero che sull'asse delle ordinate Excel mi ha messo automaticamente dei valori percentuali).




In sostanza il grafico di sinistra mi fa vedere all'incirca le stesse cose del grafico non impilato (ma in più mi dice che la popolazione totale – somma delle quattro regioni – è anch'essa cresciuta). Il grafico di destra invece mostra come sono cambiati *i rapporti reciproci* tra le regioni: ad esempio la Lombardia non solo è la regione sempre più popolata, ma l'importanza della sua popolazione, rispetto alle altre, si è ulteriormente accentuata nel corso del tempo (dal 40 a circa il 45% del totale). Si noti che quest'ultimo grafico interpreta l'esempio della ripartizione percentuale, spiegato a pag. 8.



Un risultato più piacevole a vedersi (e anche un po' più facile da leggersi) si ottiene, anziché con gli istogrammi impilati, con le "torte". Il prezzo da pagare è che ora necessariamente vediamo *un solo anno alla volta*. Vedere tutti e tre gli anni richiede di fare *tre grafici* del tutto indipendenti. (il grafico "ad anello" permetterebbe di vedere più anni su anelli concentrici, ma è di norma di difficile lettura)

Sostanzialmente nulla cambia se scegliamo gli altri tipi di grafico, quelli elencati al secondo passaggio:



Le barre sono semplicemente degli istogrammi ruotati di 90°; le linee sono degli istogrammi in cui conta solo l'altezza raggiunta, che viene raccordata tra elementi adiacenti; le aree sono delle linee "riempite". Le versioni 3D sono la copia tridimensionale di barre, istogrammi, torte e linee.

L'unico veramente diverso, e in molti casi indispensabile è il grafico a "**dispersione (XY)**". In questo caso l'asse delle X non contiene più degli elementi qualsiasi (come potevano essere le regioni), bensì è un vero asse cartesiano, disegnato in proporzione. In tutti i casi in cui non solo l'asse Y è numerico (*almeno* l'asse Y lo è sempre), ma è numerico anche l'asse X, occorre un grafico di questo tipo.

Nel nostro caso di partenza, guarda caso, gli anni sono in verità dei numeri. Noi non ce ne eravamo interessati, perché avevamo il vantaggio che l'intervallo tra essi era costante (30 anni).

Consideriamo però il caso che segue:

	A	B	C	D	E
1	Regione	1871	1931	1961	1991
2	Piemonte	2'928'160	3'457'731	3'914'250	4'290'412
3	Lombardia	3'528'732	5'595'915	7'406'152	8'831'264
4	Liguria	883'864	1'422'596	1'735'349	1'668'078
5	Veneto	2'196'208	3'487'109	3'846'562	4'363'157

Scegliamo ora il grafico a dispersione. Questa volta, per come sono messi i dati, occorre necessariamente eseguire il grafico con le serie in righe (oltre che, al solito, dire che la prima colonna è il testo della legenda).

In questo caso gli anni sono stati "spaziati" giustamente: dal 1871 al 1931 c'è il doppio che negli intervalli successivi. In più, dal momento che gli anni sono ora numeri, sull'asse Excel ha messo dei valori "in cifra tonda", di più comoda lettura.

Si noti che si potevano anche scambiare le righe con le colonne nei dati di partenza (scegliendo *Modifica, Copia*, e quindi *Modifica, Incolla speciale, Trasponi*). Si otteneva in tal caso la tabella che segue, in cui, ai fini del grafico, le serie erano ora sulle colonne.

	A	B	C	D	E
1	Regione	Piemonte	Lombardia	Liguria	Veneto
2	1871	2'928'160	3'528'732	883'864	2'196'208
3	1931	3'457'731	5'595'915	1'422'596	3'487'109
4	1961	3'914'250	7'406'152	1'735'349	3'846'562
5	1991	4'290'412	8'831'264	1'668'078	4'363'157

STAMPA ED ESPORTAZIONE DEI GRAFICI

Qualche nota riguardo alla stampa dei grafici da soli, o al loro inserimento all'interno di un documento di Word può aiutare nella risoluzione di alcuni problemi pratici.

Stampa di uno o più grafici all'interno del foglio di lavoro

Quando un foglio di lavoro (o una sua parte) viene stampato, tutti i grafici che eventualmente si trovassero sul foglio vengono stampati anch'essi, esattamente come appaiono sullo schermo (è tuttavia bene dare un'occhiata approfondita all'anteprima di stampa, perché in casi particolari potrebbero aversi lievi differenze rispetto a quanto mostrato a video).

Si fa notare che spesso è assai vantaggioso non stampare l'intero foglio, ma solo una sua parte. Per far questo, bisogna scegliere *File, Imposta pagina*, scheda *Foglio* e indicare tale parte alla casella "Area di stampa".

In alternativa è comodo usare il pulsante "Area di stampa" (se non fosse presente, lo si può aggiungere ad una barra degli strumenti, come spiegato nella nota di pag. 20).

Stampa di un grafico su un foglio a sé

Se si vuole stampare un grafico da solo, occorre selezionarlo (facendo doppio clic su di esso) e, *quando è selezionato*, scegliere *File, Anteprima di stampa*. Se i margini non sono visualizzati, renderli visibili con il pulsante "Margini". Scegliere quindi il pulsante *Imposta* e alla scheda "Grafico" selezionare "Usa pagina intera". A questo punto il grafico assumerà le dimensioni opportune in base a come vengono trascinati i margini.

Se al posto di "Usa pagina intera" si sceglie "Adatta alla pagina", il grafico verrà sì ridimensionato in base ai margini, ma mantenendo le proporzioni originali tra altezza e larghezza.

Si noti per inciso che è possibile creare un grafico su un foglio a sé, scegliendo *Inserisci, Grafico, Crea nuovo foglio*. In tal caso la stampa sarà sempre del solo grafico, nel rispetto di quanto detto sopra.

Copia e incolla di grafici da Excel a Word

In molti casi è necessario inserire i grafici all'interno di una relazione di Word. Se si seleziona il grafico, lo si copia, si va in Word e lo si incolla, in realtà non viene incollato solo il "disegno" del grafico, bensì *l'intera cartella di Excel contenente quel grafico*. Questo ha il vantaggio che il grafico sarà poi modificabile, *all'interno di Word*, semplicemente facendo doppio clic su di esso. Questa cosa è molto bella a dirsi, ma realmente utile solo in casi piuttosto particolari. Il prezzo che devo pagare per averla è quello di un sensibile appesantimento del mio file di Word: se incollo cinque grafici da una stessa cartella di Excel in un documento, quel documento finirà per contenere ben cinque copie identiche della cartella di Excel di origine.

Il più delle volte, invece, quello che voglio in Word è semplicemente un disegno del grafico – o, come si dice, un'*immagine* – che non modificherò più.

Per ottenere ciò, basta che in Word, anziché Incolla, scelga *Modifica, Incolla speciale e, appunto, "Immagine"* (l'opzione predefinita, quella che incolla l'intera cartella, sarebbe "Oggetto grafico Microsoft Excel"). Si noti che l'immagine può comunque essere ridimensionata a piacere all'interno di Word: manterrà sempre le proporzioni originali.

NOMI AL POSTO DI RIFERIMENTI

Il limite maggiore delle formule di Excel, espresse come abbiamo visto finora mediante i riferimenti, è dato dalla scarsa "leggibilità" che esse hanno, leggibilità che peggiora sensibilmente al crescere del numero di riferimenti che una formula contiene.

Per rimediare a questo problema è possibile sostituire ai riferimenti dei "**nomi**". È possibile attribuire un nome ad *una cella* o ad *un intervallo* (ivi compresa *una colonna* o *una riga intera*, o anche più colonne o righe).

In particolare, risulta assai vantaggioso definire **nomi per le colonne** in tutti i casi in cui il nostro foglio (o una sua parte) ha una caratteristica struttura tabellare *a campi e record* (come avevamo visto a pag. 10): è quello che vedremo ora.

Si può usare come nome qualunque testo non eccessivamente lungo (il massimo di 255 caratteri permessi è ... decisamente troppo!) comprendente anche punti, ma non spazi: al posto degli spazi usare le sottolineature (_). Un nome deve cominciare necessariamente con una lettera, ma poi può contenere anche numeri, purchè non assomigli ad un riferimento.

Ad esempio vanno bene: Anno1931 / Anno_1931 / Anno.1931

ma *non* vanno bene: 1931 (è un numero) / A1931 (è un riferimento) / Anno 1931 (contiene uno spazio)

I nomi si definiscono selezionando l'area che si vuole denominare e scegliendo quindi *Inserisci, Nome, Definisci*. Appare la finestra di dialogo molto semplice nella quale è possibile indicare il nome.

Il nome si digita nella prima casella. La cella o l'intervallo a cui si riferisce compare nell'ultima casella (è appunto quello della selezione attuale, ma può anche essere modificato). Si noti che esso è dato citando non solo la cella o le celle, ma anche il loro foglio ("Grafici", nell'esempio, con la consueta sintassi *NomeFoglio PuntoEsclamativo Intervallo*).

Nel nostro caso (l'esempio della volta scorsa, con le regioni e i loro abitanti), diamo ad ognuna delle tre colonne B:B, C:C e D:D il nome, rispettivamente, *Anno1931*, *Anno1961*, *Anno1991*. Notiamo che se l'intervallo attualmente selezionato ha un nome, esso compare nella casella più a sinistra della **barra della formula** ("Anno 1931" nella figura che segue, dove è selezionata la colonna B:B).

Si può usare la **casella del nome** per dare rapidamente un nuovo nome, senza passare dalla finestra di dialogo vista sopra. Basta aprire la casella a discesa e digitare il nome. Per cambiare

il riferimento di un nome esistente bisogna invece passare necessariamente dalla finestra di dialogo.

Se dalla casella a discesa si sceglie un nome esistente, viene automaticamente selezionato l'intervallo corrispondente.

Notiamo infine che, se alla cella attiva non è stato assegnato alcun nome, nella casella del nome leggiamo normalmente il suo riferimento, ad esempio *A1*. Quando però stiamo trascinando con il mouse per selezionare un intervallo, nella casella leggiamo le attuali dimensioni dell'intervallo: ad esempio *5R x 3C* (cioè 5 righe per 3 colonne): questo è spesso assai utile!

E vediamo ora come si usano i nomi. Immaginiamo di voler calcolare la variazione percentuale di abitanti tra il 1931 e il 1961 e poi tra questo e il 1991. Quello che vogliamo è dunque dato dalle ultime due colonne della tabella che segue (dove le prime quattro colonne sono i consueti dati):

	A	B	C	D	E	F
1	Regione	1931	1961	1991	Variaz. '61/31	Variaz. '91/61
2	Piemonte	3'457'731	3'914'250	4'290'412	13%	10%
3	Lombardia	5'595'915	7'406'152	8'831'264	32%	19%
4	Liguria	1'422'596	1'735'349	1'668'078	22%	-4%
5	Veneto	3'487'109	3'846'562	4'363'157	10%	13%

Usando i riferimenti, in E2 avremmo scritto:

$$=C2/B2-1$$

Avremmo poi trascinato il quadratino di trascinamento fino a E5, per completare la colonna E. Quindi, trascinando il quadratino in F2 avremmo ottenuto automaticamente la corretta formula:

$$=D2/C2-1$$

che sarebbe poi stata da trascinare nel resto della colonna F. Con i nomi, scriviamo in E2:

$$=Anno1961/Anno1931-1$$

E analogamente in F2:

$$=Anno1991/Anno1961-1$$

Poi trasciniamo entrambe le formule verso il basso nelle rimanenti righe.

Notiamo dunque i vantaggi:

- la formula "a parole" *Anno1961/Anno1931* è senza dubbio più immediatamente comprensibile che non quella con i riferimenti *C2/B2*;
- non mi devo preoccupare dei "dollari", che con i nomi non servono (concettualmente, è come se i riferimenti fossero sempre *assoluti*);
- le celle di regioni diverse (E2, E3, E4, E5) contengono tutte *l'identica formula* (non, cioè, una formula simile ma con i riferimenti aggiornati).

Il principale svantaggio è che con i nomi non posso mai contare sull'aggiornamento automatico quando trascino: essi vengono sempre copiati tal quali. Trascinando C2/B2-1 da E2 a F2 essa è diventata D2/C2-1. Con i nomi ho dovuto correggere a mano il 1961 in 1991 e il 1931 in 1961.

ULTERIORI NOTE (INTERSEZIONI DI NOMI, NOMI PREDEFINITI, NOMI LOCALI)

➤ C'è in realtà una cosa sottintesa, ma importante, che ha permesso alla nostra formula *Anno1961/Anno1931* di funzionare. I due nomi che vi compaiono si riferiscono infatti a due colonne, ma in ciascuna cella in cui ho scritto la formula viene messo, di quelle colonne, il singolo valore relativo alla regione che sto considerando.

Vale cioè questa regola: quando in una formula si usa un nome definito per un'intera colonna, Excel prende il valore dato dall'**intersezione di tale colonna con la riga in cui sto scrivendo la formula**.

Analogamente se uso un nome definito per un'intera riga, Excel prenderà il valore dato dall'intersezione di tale riga con la colonna in cui sto scrivendo la formula.

Naturalmente è possibile definire nomi relativi ad una sola cella. Se riprendo l'esempio in cui calcolavo la somma della popolazione:

	A	B	C
1	Censimenti	<i>Abitanti</i>	
2	Provincia	Censimento 1931	Censimento 1951
3	Bergamo	612'891	696'626
4	Brescia	721'890	858'243
...
13	Varese	391'449	477'055
14	LOMBARDIA	5'595'915	6'566'154

posso chiamare "Lombardia1931" la cella B14, "Lombardia1951" la cella C14, ecc., e poi usare questi nomi quando mi servo della popolazione totale per calcolare la ripartizione percentuale (pag. 8). In alternativa, posso anche chiamare "Censimento1931" l'intera colonna B:B, e analogamente le altre colonne (come ho fatto negli altri esempi di questo capitolo), e poi chiamare "Lombardia" l'intera riga 14:14. La formula della ripartizione percentuale, ad esempio per Bergamo:

$$=B3/B\$14$$

diventa in tal caso ... molto attraente:

$$=Censimento1931/Censimento1931 Lombardia$$

Il denominatore *Censimento1931 Lombardia* è proprio la cella B14, intersezione di *Censimento1931* (colonna B) con *Lombardia* (riga 14): l'**operatore di intersezione** è semplicemente *uno spazio*.

(la formula, per quanto detto sopra, funziona solo se sta sulla stessa riga della popolazione a cui si riferisce: la devo cioè scrivere sulla riga 3 per Bergamo, sulla 4 per Brescia, ecc.)

➤ Alcuni nomi vengono utilizzati da Excel per contraddistinguere aree particolari di un foglio: il più comune di questi lo abbiamo già incontrato: è l'area di stampa (pag. 27). Quando definiamo un'area di stampa non facciamo altro che definire il nome **"Area_Stampa"**. Ogni volta che Excel stampa un foglio, cerca se esiste questo nome: se lo trova, viene stampato solo l'intervallo corrispondente, in caso contrario viene stampato l'intero foglio.

Un altro nome è **"Titoli_Stampa"**, con cui si identificano una o più righe e/o una o più colonne che si vogliono stampare su ogni pagina (tipicamente, in un elenco come il nostro delle regioni, se esso fosse più lungo, tanto da non stare su una sola pagina, vorremmo ripetere la prima riga, con l'indicazione degli anni di censimento). I titoli di stampa si definiscono dalla scheda *"Foglio"* di *File, Imposta pagina*.

➤ I nomi sono di norma "visibili" da tutti i fogli di una cartella di lavoro. In alcuni casi è possibile definire nomi visibili solo da un particolare foglio. Area_Stampa è tra questi: ogni foglio può avere la propria area, e questo si ottiene definendo tale nome come **"locale"**, cioè proprio di quel foglio. Un nome viene definito come locale (nella prima casella della finestra di dialogo "Definisci nome") scrivendone la definizione nella forma *NomeFoglio PuntoEsclamativo Nome* (e infatti, se torno alla prima figura di questo capitolo, vedo che l'area di stampa compare come Grafici!Area_Stampa).

UN ESEMPIO DI FUNZIONI NIDIFICATE

Non esiste pressoché alcuna operazione che si possa immaginare di voler compiere su un foglio di calcolo, che non si riesca davvero ad ottenere, a volte con qualche "trucco", ma in genere con risultati più che accettabili.

Si pensi a questo problema pratico (è l'ultima lezione, e non posso garantire più di tanto sulla mia capacità di inventare "problemi pratici"...): si ha, da una parte, un elenco di indirizzi di alcune persone. Da un'altra parte si hanno, per ogni città, il CAP e la provincia corrispondenti. Si vuole – è facile immaginarlo – che sia Excel ad assegnare ad ogni città dell'elenco di indirizzi la provincia e il CAP letti dall'altro elenco.

Quest'ultimo è del tipo:

	A	B	C
1	<i>Comune</i>	<i>CAP</i>	<i>Provincia</i>
2	bergamo	24100	BG
3	brescia	25100	BS
4	busto arsizio	21052	VA
5	cinisello balsamo	20092	MI

Introduciamo due funzioni, che poi useremo una dentro nell'altra. Ecco la prima:

=SCARTO(rif;righe;colonne;altezza;largh)

La funzione SCARTO restituisce l'intervallo che sta, a partire dal riferimento indicato in *rif*, tante righe sotto e tante colonne a destra quanto specificato dagli argomenti *righe* e *colonne*.

L'intervallo restituito ha le stesse dimensioni di *rif*, a meno che non si specifichino dimensioni diverse mediante gli ultimi due argomenti (facoltativi) *altezza* e *largh*. In particolare, se *rif* è una cella singola, anche SCARTO restituirà una cella singola. Ad esempio:

=SCARTO(A2;3;2)

restituisce la cella C5 (3 righe sotto e due a destra di A2). Analogamente:

=SCARTO(C5;-3;-2)

restituisce A2: i numeri negativi indicano di spostarsi sopra e a sinistra.

La seconda funzione:

=CONFRONTA(valore;matrice;corrisp)

restituisce il numero di riga a cui si trova valore all'interno dell'intervallo (di norma una colonna) indicato da matrice. Ad esempio, riferendosi all'ultima tabella:

=CONFRONTA("brescia";\$A:\$A;0)

restituisce 3, perché Brescia sta sulla terza riga della colonna A (il terzo argomento deve essere sempre posto pari a 0, altrimenti la formula non funziona! Valori diversi da zero servono in casi molto più particolari, su cui non mi soffermo).

Se la tabella con CAP e provincia per ogni città sta su un foglio chiamato *Dati*, il nostro indirizzario (su un altro foglio, ad esempio "*Indirizzi*") avrà il seguente aspetto:

	A	B	C	D	E
1	Nome	Indirizzo	Città	CAP	Provincia
2	Tizio	Via Torino	Milano	=SCARTO(Dati!B\$1;CONFRONTA(\$C2;Dati!\$A:\$A;0)-1;0)	=SCARTO(Dati!C\$1;CONFRONTA(\$C2;Dati!\$A:\$A;0)-1;0)
3	Rossi	Piazza Garibaldi	Vigevano	=SCARTO(Dati!B\$1;CONFRONTA(\$C3;Dati!\$A:\$A;0)-1;0)	=SCARTO(Dati!C\$1;CONFRONTA(\$C3;Dati!\$A:\$A;0)-1;0)
4	Dante	Viale Repubblica	Pavia	=SCARTO(Dati!B\$1;CONFRONTA(\$C4;Dati!\$A:\$A;0)-1;0)	=SCARTO(Dati!C\$1;CONFRONTA(\$C4;Dati!\$A:\$A;0)-1;0)

Vediamo la formula del CAP, per il primo indirizzo:

=SCARTO(Dati!B\$1;CONFRONTA(\$C2;;0)-1;1)

Dalla cella B1 del foglio Dati mi sposto 0 colonne a destra (ultimo 0 della funzione), perché il CAP, nel foglio Dati, è proprio sulla colonna B. E quante righe sotto? Quelle date dalla funzione CONFRONTA, che cerca il valore C2 (Milano) nella colonna Dati!\$A:\$A.

In realtà, devo fare attenzione a scrivere CONFRONTA(.)-1: devo sottrarre una riga (il -1) perché CONFRONTA restituisce il numero assoluto della riga a cui trova Milano, mentre SCARTO vuole quante righe Milano sta sotto la cella A1: se Milano sta, ad esempio, alla quarta riga, significa che è tre righe sotto A1!

Un'occhiata ai dollari: \$C2 mi resta C quando passo nella colonna di destra a calcolare la provincia, ma mi diventa 3, 4, ecc. quando scendo agli indirizzi successivi. Dati!\$A:\$A deve sempre rimanere tale, perché è la colonna con i nomi delle città. Dati!B\$1 infine è sicuramente assoluto come riga (\$1), perché lo scarto lo voglio sempre dalla prima riga; l'ho messo relativo come colonna, perché così mi diventa C\$1 per la provincia, che è subito a destra del CAP sia nei dati, sia negli indirizzi (è un piccolo trucco, potrei anche non badarci, fare \$B\$1 e correggere a mano per la provincia).

La stessa formula ora con i nomi:

Chiamo:

Comune Dati!\$A:\$A

MioComune Indirizzi!\$C\$C

ed ecco le due funzioni, per CAP e provincia:

=SCARTO(Comune;CONFRONTA(MioComune;Comune;0)-1;1;1;1)

=SCARTO(Comune;CONFRONTA(MioComune;Comune;0)-1;2;1;1)

In questo caso:

- parto sempre da Comune (1° argomento di scarto): il CAP gli è (nel foglio Dati) una colonna a destra, la provincia due colonne a destra (rispettivamente 1 e 2 come terzultimo argomento delle due formule);
- devo aggiungere 1;1, quarto e quinto argomento di SCARTO (cioè *altezza* e *largh*) per dire che voglio comunque *una singola cella*, anche se *rif* (primo argomento di SCARTO) è un'intera colonna. Senza questo dettaglio la formula non funzionerebbe.

C'è ancora una cosa che non mi piace (ma giusto perché voglio esagerare!): devo contare a mano quante colonne a destra di Comune si trovano (nel foglio Dati) il CAP e la provincia. Ma esiste la funzione:

=RIF.COLONNA(rif)

che mi dà il numero della colonna a cui si trova *rif*. E allora:

RIF.COLONNA(Cap)-RIF.COLONNA(Comune)

è proprio quante colonne a destra di Comune si trova il CAP.

Le due formule nella versione "elegantissima" sono dunque:

=SCARTO(Comune;**CONFRONTA(MioComune;Comune;0)**-1;RIF.COLONNA(Cap)-
RIF.COLONNA(Comune);1;1)

=SCARTO(Comune;**CONFRONTA(MioComune;Comune;0)**-1;RIF.COLONNA(Provincia)-
RIF.COLONNA(Comune);1;1)

dove con RIF.COLONNA ho sostituito i valori 1 e 2 terzultimo argomento di SCARTO. Naturalmente ho dovuto definire i nuovi nomi:

CAP Dati!\$B:\$B

Provincia Dati!\$C:\$C

ma adesso potrei spostare a piacere le loro colonne, che la formula continuerebbe a funzionare.

Un'ultimissima nota: se gli elenchi fossero molto grossi (centinaia di città in Dati e centinaia di persone in Indirizzi), le formule diventerebbero un po' pesanti per Excel da ricalcolare: varrebbe la pena, se si immagina che i dati non cambiano, di sostituirle con i corrispondenti valori, una volta calcolate la prima volta. Basta copiarle, e poi scegliere *Modifica, Incolla speciale, Valori*, come già visto a pag. 16.

SUGGERIMENTI PER UN MIGLIORE USO DI EXCEL

TASTI DI SCELTA RAPIDA

- F2 Quando una qualsiasi cella è selezionata, permette di modificarne il contenuto nella barra della formula. È di norma molto più veloce che non fare clic con il mouse sulla barra della formula o fare doppio clic sulla cella. Si ricorda che la **"barra della formula"** è la lunga casella immediatamente sotto le barre degli strumenti, in cui compare appunto la formula della cella attiva.
- F4 Quando si sta scrivendo o modificando un riferimento in una formula, alterna i riferimenti assoluti e relativi (mette e toglie i dollari, vedi pag. 7. In tutti gli altri casi, ripete l'ultima azione che è stata fatta. Ad esempio se ho messo in grassetto una cella, mi sposto su un'altra e premo F4, anche l'altra diventa in grassetto. L'uso come "ripeti" vale anche in Word.
- F9 Ricalcola i fogli di lavoro quando è impostato il calcolo manuale (vedi sotto, tra le "opzioni utili")
- Ctrl X, C, V Sono i tasti di scelta rapida rispettivamente di taglia, Copia e incolla. (vedi a pag. 14). Valgono anche in Word e pressoché in tutte le altre applicazioni.
- Ctrl G, I, S Sono i tasti di scelta rapida rispettivamente di grassetto, corsivo (la 'i' sta per Italo) e sottolineato. Valgono anche in Word e in molti altri programmi di scrittura.
- Shift Ctrl T, S Sono i tasti di scelta rapida rispettivamente per Trova e Sostituisci. Valgono anche in Word. Alt backspace, Ctrl Z Entrambi i tasti annullano l'ultima azione. Valgono anche in Word e in molte altre applicazioni

Una chicca per te che ti permetterà di risparmiare tanto tempo di battitura:

- Shift F3 Quando si vuole modificare l'impostazione delle maiuscole di una parola. Ad esempio se utilizzo la combinazione su una parola ad esempio "ciao" la stessa diventerà "Ciao" con la prima lettera maiuscola. Se uso la combinazione una seconda volta diventerà "CIAO" tutto maiuscolo.

OPZIONI UTILI

Elenco qui un insieme di opzioni che di norma non vanno bene come sono predefinite, o che comunque è utile sapere di poter cambiare. Esse si trovano in *Strumenti, Opzioni* e quindi nelle relative schede che indico caso per caso.

Modifica direttamente nella cella: Se si disattiva questa opzione, non si può più modificare una cella facendo doppio clic su di essa (ma solo facendo clic sulla barra della formula o premendo F2). In compenso però, se la cella contiene una funzione, con doppio clic *seleziono tutte le celle citate da quella funzione*. Questa possibilità è davvero utilissima per controllare a colpo d'occhio se ho scritto i riferimenti giusti.

Calcolo manuale: In alcuni casi (fogli con molte formule molto complesse), può essere utile o indispensabile impostare il **calcolo manuale**, in modo che il ricalcolo delle celle non avviene come di consueto automaticamente, ma solo quando lo si richiede esplicitamente, premendo il tasto F9. Il calcolo manuale si imposta dalla scheda *Calcolo*.

Visualizza formule: Se il foglio contiene molte formule e se ne si vuole avere una panoramica, può essere utile visualizzare *temporaneamente* in tutte le celle le formule anziché i normali valori. L'opzione si trova nella scheda *Visualizza* e va poi disattivata per tornare a vedere il foglio nella maniera consueta.

Fogli nella nuova cartella: Spesso non è necessario avere decine di fogli vuoti in una nuova cartella di lavoro. Impostare questa opzione a 1 e aggiungere i fogli solo quando servono, mediante *Inserisci, Foglio di lavoro*. L'opzione si trova nella scheda *Generale*.

Posizione file predefinito: Per fare in modo che le finestre "Apri" e "Salva con nome" si posizionino automaticamente su una directory predefinita, indicarla qui (ad esempio C:\MioExcel). L'opzione si trova nella scheda *Generale*.

Inserimento ed eliminazione dinamici: a partire da Excel 95, il computer tenta di muovere i fogli non come di consueto a scatti, ma in modo "dinamico", con l'effetto di un ... "mal di mare" assolutamente garantito: disattivare pertanto l'opzione (scheda *Modifica*)

Stampa intestazioni di riga e colonna: è un'opzione di stampa (si trova pertanto in *File, Imposta Pagina, Foglio*), che permette di stampare anche le lettere che indicano le colonne e i numeri che indicano le righe. Di solito non serve, ma può essere molto utile se si devono controllare delle formule, magari confrontando un foglio stampato con una schermata a video.